

EDITORALUTÉCIA

PROGRAMAS DE JOGOS DE ESPIONAGEM



Para o Spectrum, TRS-80, TRS-80 Color, Apple, MSX, PC, Commodore 64,

VIC 20 e todos os micros nacionais compatíveis.

Titulo original inglês
COMPUTER SPY GAMES

Copyright © 1984 by Usborne Publishing Ltd.

Direitos de publicação exclusiva em língua portuguesa em todo o mundo
adquiridos pela

EDITORA LUTÉCIA LTDA.

Rua Argentina 171 — 20921 Rio de Janeiro, RJ — Tel.: 580-3668
que se reserva a propriedade literária desta tradução

Impresso no Brasil

PEDIDOS PELO REEMBOLSO POSTAL
Caixa Postal 23.052 — Rio de Janeiro, RJ — 20922

Impresso por
Gráfica Portinho Cavalcanti Ltda.
Rua Santana, 136/138 (edifício próprio)
Tel.: 224-7732 (PABX)
Rio de Janeiro — RJ

PROGRAMAS DE JOGOS DE ESPIONAGEM

Jenny Tyler e Chris Oxlade

Tradução e Adaptação de Ronaldo Sergio de Biasi



Sumário

- 2 Introdução
- 3 Olhos de Espião
- 4 Holofote
- 6 Espiobot
- 8 Teste para Espiões
- 10 Gerador de Mensagens Secretas
- 12 Encontro Marcado
- 16 Código Morse
- 17 Respostas dos Problemas

EDITORAL LUTÉCIA

Introdução

Embora todos os micros nacionais utilizem a linguagem BASIC, existem vários dialetos desta linguagem usados em micros de diferentes marcas e modelos. Os micros nacionais podem ser divididos em seis "famílias": a família Spectrum (TK-85, TK-90X), a família TRS-80 (DGT-1000, CP-500 etc.), a família TRS-80 Color (CP-400), a família Apple (Micro Engenho, Ap-II etc.), a família MSX (Expert, Hot Bit) e a família PC (Ego, I-7000 etc.). Se você não souber a que família pertence o seu micro, poderá obter a informação em qualquer loja especializada.

Neste livro, a listagem principal de cada programa é para os micros da família TRS-80. As linhas que precisam ser mudadas para que o programa funcione em outros computadores estão indicadas por símbolos e as modificações aparecem no final de cada listagem. Os símbolos são os seguintes:

- família TRS-80 Color
- ▼ família Spectrum
- família Apple
- família MSX
- ◆ família PC
- ▲ Commodore 64 e VIC 20*

Cada vez que você vir o símbolo da família a que pertence o seu micro, procure no final da listagem uma linha com o mesmo símbolo e o mesmo número. Use-a no lugar da linha correspondente da listagem principal. Na página 17 há notas especiais para os usuários do TK-2000, do MSX e do PC.

*Até o momento não têm equivalentes no Brasil.

Como carregar e executar os programas

Escreva as linhas exatamente como aparecem na listagem, incluindo todos os espaços e sinais de pontuação. Depois de bater cada linha, verifique na tela se tudo está certo antes de apertar a tecla RETURN (ou a tecla equivalente para o seu micro).

Para começar o jogo, entre com RUN. Em alguns jogos, as coisas acontecem muito depressa, de modo que é melhor ler atentamente as instruções antes de começar. Se o programa não funcionar corretamente, o melhor é listá-lo e comparar as instruções com as do livro, uma por uma, até descobrir onde está o erro.

Experiências com os jogos

Este livro contém várias sugestões para modificar e ampliar os programas, mas não tenha medo de introduzir suas próprias alterações. Não há perigo de danificar o computador; se sua idéia não der certo, você sempre poderá voltar ao programa original.


Você provavelmente terá vontade de mudar a velocidade de alguns jogos, especialmente depois que estiver familiarizado com eles. As instruções para fazer isso estão perto da listagem de cada programa.

Se o seu computador é capaz de gerar cores ou sons, consulte o manual para descobrir como funcionam as instruções correspondentes e experimente usá-las nos programas deste livro.

Ilustrações: Sue Walliker, Rob McCaig e Martin Newton.

Diagramação: Roger Priddy.

Programa "Espibot": Adrian Hall.



Olhos de Espião

Se você acha que daria um bom espião, experimente este jogo.

O computador mostra na tela os números de 1 a 9. Examine atentamente a posição dos números antes de apertar uma tecla (qualquer tecla serve). Um dos números muda de posição. Quais deles? Quando achar que sabe a resposta, aperte de novo uma tecla e entre com o seu palpite. Vai ver que seus poderes de observação não são tão fabulosos quanto pensava.

Como o programa funciona

- 10: Reserva espaço na memória para as posições dos números.
- 20: Inicia a contagem de pontos em zero.
- 30-60: Escolhe as posições dos números.
- 70: Vai para a sub-rotina que mostra os números na tela.
- 80: Vai para a sub-rotina que espera até que uma tecla seja apertada.
- 90-100: Vai para a sub-rotina que escolhe um dos números e o chama de M.
- 110: Decide se o número vai ser movido para a direita ou para a esquerda.
- 120: Vai para a sub-rotina que mostra os números na tela.
- 130: Vai para a sub-rotina que espera até que uma tecla seja apertada.
- 140-150: Limpa a tela e imprime mensagem.
- 160: Recebe um número.
- 170: Se o número está errado, pula para as mensagens das linhas 250 a 280.
- 180-190: Limpa a tela e imprime mensagem.
- 200-210: Aumenta a contagem de pontos e mostra o resultado.
- 220-230: Imprime mensagem e vai para a sub-rotina que espera até que uma tecla seja apertada.
- 240: Volta para uma nova tentativa.
- 250-300: Informa que você errou e pergunta se quer jogar de novo.
- 310-330: Sub-rotina que espera até que uma tecla seja apertada.
- 340-350: Sub-rotina para escolher um número ao acaso entre 1 e 9.
- 360-400: Sub-rotina para mostrar os números na tela, nas posições escolhidas.

```

10 DIM X(9):DIM Y(9)
20 LET P=0
30 FOR I=1 TO 9
40 GOSUB 340:LET X(I)=N+3
50 GOSUB 340:LET Y(I)=N+3
60 NEXT I
70 GOSUB 360
80 GOSUB 310
90 GOSUB 340
100 LET M=N:GOSUB 340
110 LET X(M)=X(M)+SGN(N-5.1)
120 GOSUB 360
130 GOSUB 310
140 CLS:PRINT
150 PRINT "QUAL O NUMERO QUE SE MOVEU?"
160 INPUT A
170 IF A<>M THEN GOTO 250
180 CLS:PRINT
190 PRINT "MUITO BEM!"
200 LET P=P+1
210 PRINT "AGORA VOCE TEM ";P;" PONTOS"
220 PRINT:PRINT "APORTE UMA TECLA"
230 GOSUB 310
240 GOTO 30
250 CLS:PRINT:PRINT "ERRADO...FIM DO JOGO"
260 PRINT "A RESPOSTA CERTA ERA ";M
270 PRINT "VOCE FEZ ";P;" PONTOS"
280 PRINT "QUER JOGAR DE NOVO? (S/N)"
290 INPUT AS:IF AS="S" THEN RUN
300 STOP
310 LET IS=INKEY$
320 IF IS="" THEN GOTO 310
330 RETURN
340 LET N=INT(RND(0)*9)+1
350 RETURN
360 CLS
370 FOR I=1 TO 9
380 PRINT@ Y(I)*64+X(I),STR$(I);
390 NEXT I
400 RETURN


```

Modificações

- 140,180,250,360 Substitua CLS por HOME
- ▲ 140,180,250,360 Substitua CLS por PRINT CHR\$(147)
- 310 IS="" : IF PEEK(-16384)>127 THEN GET IS
- ▲ 310 GET IS
- ▲ 340 LET N=INT(RND(1)*9+1)
- ▼ 340 LET N=INT(RND*9+1)
- 380 PRINT@ Y(I)*32+X(I),STR\$(I);
- 380 VTAB(Y(I)):HTAB(X(I)):PRINT STR\$(I)
- ▼ 380 PRINT AT Y(I),X(I);STR\$(I)
- ▲ 380 PRINT CHR\$(19):FOR LL=1 TO Y(I):PRINT:NEXT LL
- PRINT TAB(X(I));STR\$(I)
- ◆ 380 LOCATE X(I),Y(I):PRINT STR\$(I)



Veja se é capaz de modificar o programa para que ele mostre o recorde do momento.



Esta noite você tem que executar uma missão extremamente perigosa: atravessar uma região muito vigiada do território inimigo e voltar, esquivando-se aos raios de um potente holofote. Você pode se esconder atrás de árvores e pedras, mas se for visto... não terá uma segunda oportunidade. Depois de completar uma missão, receberá outra, ainda mais difícil. Quantas missões será capaz de cumprir antes de ser descoberto pelo inimigo?

Use a tecla M para andar para a direita e N para andar para a esquerda. Para completar uma missão, você precisa cruzar o terreno duas vezes, uma na ida e outra na volta.

```

10 GOSUB 450
20 LET A=1:LET G=0:LET S=0
▲ 30 CLS
40 LET X=0:LET Y=12:LET B$=A$(A)
50 GOSUB 380
60 LET F=0:LET N=0:LET NN=0:GOSUB 340
70 LET L=0:LET C=0:LET TC=10:LET C1=0
▲ 80 LET I$=INKEY$
90 IF I$="N" THEN LET NN=N-1
100 IF I$="M" THEN LET NN=N+1
110 IF NN>19 THEN LET NN=19
120 IF NN<0 THEN LET NN=0
130 IF NN=19 AND F=0 THEN LET F=1
140 IF NN=0 AND F=1 THEN LET F=2
150 GOSUB 340
160 IF N<>NN THEN LET S=S+1
170 LET N=NN:LET G=G+1
180 GOSUB 400
▼ 190 IF MID$(A$(A),N+1,1)=" " AND L=1 THEN GOTO 240
200 FOR T=1 TO 50:NEXT T
210 IF F<2 THEN GOTO 80
    
```

Você sabe como fazer o holofote ficar aceso por mais tempo?

Talvez você possa substituir os obstáculos, indicados nesta listagem pelo símbolo =, por símbolos gráficos que representem árvores, construções etc.



Como o programa funciona

- 10: Vai para a sub-rotina que lê os dados.
- 20: Dá valores iniciais às variáveis.
- 40-50: Imprime os obstáculos.
- 60-70: N é a posição do jogador. NN é a nova posição.
- 80: Verifica se uma tecla está sendo apertada.
- 90-100: Se a tecla apertada é N (esquerda), a nova posição é N menos 1. Se é M (direita), a nova posição é N mais 1.
- 110-120: Evita que o jogador saia da tela.
- 130-140: F se torna igual a 1 quando o jogador chega à extremidade direita da tela e igual a 2 quando ele chega de novo à extremidade esquerda.
- 150: Vai para a sub-rotina que muda a posição do jogador.
- 160: Quando o jogador se move, S aumenta.
- 170: Faz a posição do jogador igual à nova posição e aumenta G. G funciona como um contador de tempo, aumentando sempre, mesmo que o jogador não se mova.
- 180: Vai para a sub-rotina que acende e apaga o holofote.
- 190: Verifica se o jogador está abaixo de um espaço vazio com o holofote ligado; em caso afirmativo, vai para o fim do jogo.
- 200: Pausa.
- 210: Se o jogador não voltou ao lado esquerdo da tela, verifica de novo se alguma tecla está sendo apertada.

Faça soar um
alarme quando o
jogador for visto.

```
220 LET A=A+1:IF A=8 THEN LET A=7
230 GOTO 30
240 LET X=4:LET Y=1:LET BS="VOCE FOI VISTO"
250 GOSUB 380:PRINT
260 PRINT "VOCE FEZ ";INT((A-1+S/G)*100)
270 PRINT:PRINT "QUER JOGAR DE NOVO? (S/N)"
280 INPUT CS:IF CS="S" THEN RUN
290 PRINT "TCHAU...":STOP
300 LET Y=3:LET X=10:LET BS="*"
310 GOSUB 380:RETURN
320 LET X=10:LET Y=3:LET BS=" "
330 GOSUB 380:RETURN
340 LET X=N:LET Y=13:LET BS=" "
350 GOSUB 380
360 LET X=NN:LET BS="S"
370 GOSUB 380:RETURN
```

```
▲●● 380 PRINT@ Y*64+X,BS;
390 RETURN
400 IF L=1 THEN LET C=C+1
◆ 410 IF C=TC THEN LET L=0:LET C=0:LET TC=INT(RND(0)*8+(12-A)):GOSUB 320
420 IF L=0 THEN LET C1=C+1
◆ 430 IF C1=TC THEN LET L=1:LET C1=0:LET TC=INT(RND(0)*10+(8-A)):GOSUB 300
440 RETURN
▼ 450 DIM AS(7)
460 FOR I=1 TO 7:READ AS(I):NEXT I
470 RETURN
480 DATA "=="
490 DATA "=="
500 DATA "=="
510 DATA "=="
520 DATA "=="
530 DATA "=="
540 DATA "=="
```

Você pode mudar
a posição dos
obstáculos
mudando as linhas
de dados.

220: Aumenta o nível de dificuldade.
230: Volta para imprimir novos
obstáculos.

240-290: Imprime mensagem, a
contagem final e pergunta se quer
jogar de novo.

300-310: Imprime o holofote.

320-330: Apaga o holofote.

340-370: Sub-rotina que muda a
posição do jogador.

380-390: Sub-rotina que imprime
símbolos na tela a partir das
coordenadas X,Y.

400-440: Sub-rotina que escolhe o
tempo que a luz vai passar ligada e o
tempo que vai passar desligada.

450-470: Sub-rotina que lê os dados.


480-540: Linhas de dados.

Como você faria para dar
ao jogador uma segunda
oportunidade, tornando
assim o jogo mais fácil?

Modificações

```
● 30 HOME
▲ 30 PRINT CHR$(147)
▲ 80 GET IS
● 80 IS="":IF PEEK(-16384))127 THEN GET IS
▼ 190 IF AS(A,N+1)=" " AND L=1 THEN GOTO 240
○ 380 PRINT@ Y*32+X,BS;
▼ 380 PRINT AT Y,X,BS
● 380 VTab(Y):HTab(X+1):PRINT BS
▲ 380 PRINT CHR$(19):FOR LL=1 TO Y:PRINT:NEXT:PRINT TAB(X);BS
◆ 380 LOCATE X,Y:PRINT BS
▼ 410,430 Substitua RND(0) por RND
▲●● 410,430 Substitua RND(0) por RND(1)
▼ 450 DIM AS(7,20)
```

Espiobot



Você foi encarregado de controlar o Espiobot, um robô operado à distância que é capaz de seguir agentes inimigos. Você é informado a respeito dos movimentos do agente nas ruas da cidade (se ele dobrou à direita ou à esquerda) e deve transmitir a mesma informação ao Espiobot, para que ele não perca o agente de vista.

Infelizmente, o agente sabe que está sendo seguido; à medida que o tempo passa, torna a sua tarefa cada vez mais difícil.

Use a tecla D para informar ao Espiobot que o agente foi para a direita e E para informar que foi para a esquerda. Por quanto tempo conseguirá seguir o agente inimigo?

Você sabe como fazer o computador perguntar o nome do jogador se ele bater o recorde?



```

10 LET SP=0.3
▲● 20 CLS
30 LET H=0
40 LET K=0:LET S=0:LET L=1:PRINT
50 PRINT ">>> ESPIOBOT <<<":PRINT
60 PRINT "APERTE P PARA PARTIR"
▲● 70 LET CS=INKEY$
80 IF CS<>"P" THEN GOTO 70
▲● 90 CLS:PRINT:PRINT
100 IF K=5 THEN LET L=L+1:LET K=0
110 LET MS=""
120 LET CS=""
130 FOR T=1 TO L
◆▼▲● 140 LET N=RND(0)
150 IF N>0.5 THEN PRINT "ESQ":LET MS=MS+"E"
160 IF N<0.5 THEN PRINT "DIR":LET MS=MS+"D"
170 NEXT T
180 LET K=K+1
190 FOR D=1 TO 200*L*SP
▲● 200 LET CS=INKEY$
210 NEXT D
▲● 220 CLS:PRINT:PRINT
230 LET E=0
240 PRINT "DIGA AS DIRECOES"
250 FOR T=1 TO L
260 PRINT "?";
▲● 270 LET CS=INKEY$
280 IF CS<>"D" AND CS<>"E" THEN GOTO 270
290 PRINT CS
▼ 300 IF CS<>MID$(MS,T,1) THEN LET E=1
310 NEXT T
320 IF E=0 THEN LET S=S+L
330 IF S>H THEN LET H=S
340 IF E=1 THEN PRINT "VOCE O DEIXOU ESCAPAR!"
350 PRINT "VOCE FEZ ";S;" PONTOS"
360 PRINT "O RECORDE E ";H;" PONTOS"
370 IF E=1 THEN GOTO 40
380 FOR D=1 TO 500*SP
390 NEXT D
400 GOTO 90

```

Modificações

- 20,90,220 Substitua CLS por HOME
- ▲ 20,90,220 Substitua CLS por PRINT CHR\$(147)
- 70,200,270 CS="" : IF PEEK(-16384)>127 THEN GET CS
- ▲ 70,200,270 GET CS
- ◆▼ 140 LET N=RND
- ▲● 140 LET N=RND(1)
- ▼ 300 IF CS<>MS(T) THEN LET E=1

Como funciona

30-40: Dá valores iniciais às variáveis. L é o número de palavras que aparecem na tela. K indica quantas vezes apareceram L palavras.

60-80: Imprime mensagem e espera que a tecla P seja apertada para começar.

90: Limpa a tela e deixa duas linhas em branco.

100: Se K é 5, aumenta L e faz K igual a zero.

110-120: Inicializa duas variáveis "string".

130-170: Escolhe ao acaso uma sequência de L direções, mostra essas direções na tela e as guarda em MS.

180: Aumenta o valor de K.

190-210: Pausa para que o jogador possa ver as direções. A duração aumenta com o número de palavras.

230: Indicador de erro; se o jogador comete um engano, E passa a ser igual a 1.

250: Começo de um "loop" para receber e verificar as respostas.

260-270: Imprime um ponto de interrogação, recebe uma direção e a guarda em CS.

280: Se a tecla apertada não é D nem E, continua esperando.

300: Verifica se a letra em CS corresponde à letra apropriada em MS; caso não corresponda, faz E igual a 1.

320: Aumenta a contagem de pontos se não houve erros.

330: Anota o novo recorde se necessário.

340: Imprime mensagem em caso de erro.

350-360: Imprime a contagem e o recorde.

370: Volta para um novo jogo.

380-390: Pausa para mostrar a contagem.

400: Volta para uma nova rodada.

A velocidade do jogo é controlada pela linha 10. Para começar, use os seguintes valores: Apple 0.1, Spectrum 0.2, TRS-80/TRS-80 Color/MSX 0.3, Commodore 64/VIC 20 0.4, PC 0.5. Independentemente do tipo de micro, quanto menor o número, mais rápido será o jogo.



Teste para Espiões

Como aluno da Escola de Espionagem, você tem que passar em uma série de testes para subir de posição. Você começa no nível mais baixo, o de Espião Aprendiz Nível 5. Seu objetivo é chegar ao topo e tornar-se um EMI* Nível 1, ou mesmo conquistar o grau máximo de Superespião.

Em cada teste, você dispõe de dez posições na tela do computador e recebe uma série de números entre 1 e 99. Seu objetivo é colocar esses números em ordem nas dez posições, com o menor na posição 1 e o maior na posição 10.

Você tem direito a descartar alguns números que não sirvam apertando a tecla D. O número de números que você pode descartar é igual ao número do nível onde você está.

Como o programa funciona

```
▼10 DIM N(10):DIM NS(5)
▲●20 CLS
30 GOSUB 510
40 LET WS=""
50 LET D=5
60 LET G=0
70 FOR I=1 TO 10:LET N(I)=0:NEXT I
80 LET I=1
90 GOSUB 430
♦VANO 100 LET M=INT(RND(0)*99+1)
110 PRINT:PRINT "ONDE VOCE COLOCA ";M;" ?"
120 PRINT:INPUT PS
130 IF PS="D" AND G<D THEN LET G=G+1:GOTO 90
140 IF PS="D" THEN PRINT "NAO PODE!":GOTO 120
150 LET P=VAL(PS)
160 IF P<1 OR P>10 THEN GOTO 120
```

10: Reserva espaço na memória para as matrizes.

20: Limpa a tela.

30: Vai para a sub-rotina que lê os dados a serem guardados na matriz NS.

40: Cria uma "string" vazia.

50: Começa no nível 5.

60: G é o número de números descartados.

70: Coloca zeros nas 10 posições para começar.

80: Conta os números que foram colocados.

90: Vai para a sub-rotina que imprime o seu nível, os números de 1 a 10 e os números que já foram colocados.

100: Escolhe um número entre 1 e 99.

110-120: Imprime o número e pergunta onde você quer colocá-lo.

130-140: Verifica se você apertou D e se ainda tem direito de descartar números.

150: Calcula o valor numérico da tecla que você apertou.

160: Se a tecla apertada não está entre 1 e 10, continua esperando.

Você sabe como mudar o número de posições de 10 para 15?



Mude os nomes dos níveis se quiser.

*Espião Muito Importante.

Chegarei a Superespião?




```

170 IF N(P)>0 THEN PRINT "ESTA OCUPADO":GOTO 120
180 LET N(P)=M
190 LET F=0
200 FOR L=P TO 10
210 IF N(L)<M AND N(L)<>0 THEN LET F=1
220 NEXT L
230 FOR L=1 TO P
240 IF N(L)>M AND N(L)<>0 THEN LET F=1
250 NEXT L
260 IF F=1 THEN GOTO 360
270 LET I=I+1:IF I<11 THEN GOTO 90
280 LET D=D-1:IF D=0 THEN GOTO 330
290 PRINT "MUITO BEM. PASSOU PARA NIVEL ";D
300 PRINT:PRINT "VOCE AGORA E UM ";N$(D)
310 LET W$=""
320 GOTO 400
330 PRINT "PARABENS! VOCE CHEGOU AO NIVEL"
340 PRINT "DE SUPERESPIAO!"
350 STOP
360 PRINT "ERRADO. SINTO MUITO."
370 PRINT:PRINT "VOCE AINDA E UM ";
380 PRINT N$(D)
390 LET W$="AINDA"
400 PRINT:PRINT "QUER TENTAR DE NOVO? (S/N)"
410 INPUT A$:IF A$="S" THEN GOTO 60
420 STOP
▲● 430 CLS
440 PRINT:PRINT "VOCE E ";W$;" UM ";N$(D)
450 PRINT
460 FOR J=1 TO 10
470 PRINT J;
480 IF N(J)>0 THEN PRINT N(J);
490 PRINT:NEXT J
500 RETURN
510 FOR I=1 TO 5:READ N$(I)
520 NEXT I
530 RETURN
540 DATA "EMI","ESPIAO","ESPIAO JUNIOR"
550 DATA "ESPIAO ASSISTENTE","ESPIAO APRENDIZ"

```

170: Verifica se a posição já está ocupada e informa ao jogador.

180: Coloca o número na posição indicada.

190-220: Verifica todas as posições abaixo da indicada para ver se alguma contém um número menor. Em caso afirmativo, faz F igual a 1.

230-250: Verifica todas as posições acima da indicada para ver se alguma contém um número maior. Em caso afirmativo, faz F igual a 1.

260: Se a posição está errada, pula para a linha 360.

270: Aumenta o contador e volta para escolher outro número.

280-300: Muda D para o nível seguinte e imprime mensagem (se D é 0, vai para a mensagem de Superespião).

310: Coloca uma "string" vazia em W\$.

320: Pula para a mensagem "Quer tentar de novo?"

330-350: Mensagem de Superespião.

360-380: Mensagem de erro.

390: Coloca a palavra "ainda" em W\$.

400-420: Pergunta se você quer jogar de novo; se a resposta é não, encerra o jogo.

430-500: Sub-rotina que imprime o nível e os números.

510-530: Sub-rotina que coloca os nomes dos níveis na matriz N\$.

540-550: Dados.

Modificações

- ▼ 10 DIM N(10):DIM N\$(5,16)
- 20,430 Substitua CLS por HOME
- ▲ 20,430 Substitua CLS por PRINT CHR\$(147)
- ◆ 100 LET M=INT(RND*99+1)
- ▼ 100 LET M=INT(RND(1)*99+1)



Gerador de Mensagens Secretas

Use este programa para enviar mensagens em código a seus amigos. Eles vão precisar de um computador para decodificar as mensagens, mas não necessariamente do mesmo tipo que o seu. (É claro que também vão precisar de uma cópia do programa.) Eles também poderão mandar mensagens em código para você.



Você e seus amigos
poderão formar uma Rede
de Espionagem.

```

▲● 10 CLS:PRINT:PRINT "MENSAGEM SECRETA"
20 PRINT "=====
30 PRINT:PRINT "VOCE QUER:"
40 PRINT
50 PRINT " 1.CODIFICAR UMA MENSAGEM"
60 PRINT "OU 2.DECODIFICAR UMA MENSAGEM"
70 PRINT:PRINT
80 PRINT "ENTRE COM O NUMERO":INPUT A
90 IF A=1 THEN GOSUB 120
100 IF A=2 THEN GOSUB 210
110 GOTO 30
120 LET CS="CODIFICADA":GOSUB 400
▼ 130 LET X=INT(RND(0)*25+1)
140 LET MS="F"+MS
150 GOSUB 310:GOSUB 420
160 LET MS=CHR$(X+64)+MS
170 IF LEN(MS)/2=INT(LEN(MS)/2) THEN GOSUB 450
180 PRINT "MENSAGEM EM CODIGO:"
190 PRINT MS
200 RETURN
210 LET CS="DECODEIFICADA":GOSUB 400
220 IF LEN(MS)/2=INT(LEN(MS)/2) THEN GOSUB 450
▼ 230 LET KS=LEFT$(MS,1)
▼ 240 LET MS=RIGHT$(MS,LEN(MS)-1)
▼ 250 LET X=ASC(KS)-64
260 LET X=-X:GOSUB 420
    
```

Como o programa funciona

10-80: Imprime o "menu" e recebe sua opção.

90-100: Vai para a sub-rotina apropriada.

110: Se a escolha não faz sentido, volta para o menu.

120-200: SUB-ROTINA PARA CODIFICAR

120: Vai para a sub-rotina que recebe a mensagem.

130: Escolhe de quanto as letras vão ser deslocadas e guarda este valor em X.

140: Acrescenta uma letra para que, caso a primeira palavra tenha apenas uma letra, a mensagem codificada não comece com um espaço.

150: Vai para duas sub-rotinas, uma que desloca a mensagem de X letras e outra que a inverte.

160: Acrescenta uma letra à mensagem para informar à sub-rotina de decodificação qual é o valor de X.

170: Verifica se a mensagem tem um número par de letras. Em caso afirmativo, vai para uma sub-rotina que troca as letras duas a duas.

210-300: SUB-ROTINA PARA DECODIFICAR

210: Vai para a sub-rotina que recebe a mensagem.

220: Se a mensagem tem um número par de letras, vai para uma sub-rotina que troca as letras duas a duas.

230-240: Suprime a primeira letra. (Esta é a letra que foi acrescentada na linha 160 para indicar o deslocamento.)

250: Calcula o valor de X.

260: Muda o sentido do deslocamento e vai para a sub-rotina que inverte a mensagem.

RSKSU SZFAE EW GSLKW
KGFS DH KGX

XCAMT KMA
PYJD PCSO
CDCFA MD

EDPZWMS Y VSXYET
GKSQMRM
MJS

RGLFSMIFW JGH SD
JSUAX EWVGHX

KJXRWK KOSSBO
AEK SQPN

QBKZDCG CNHH




```

▼ 270 GOSUB 310:M$=RIGHT$(M$,LEN(M$)-1)
280 PRINT "MENSAGEM DECODIFICADA:"
290 PRINT M$
300 RETURN
310 LET N$=""
320 FOR I=1 TO LEN(M$)
▼ 330 LET Q$=MID$(M$,I,1):LET N=ASC(Q$)
340 IF N=32 THEN GOTO 380
350 LET N=N+X
360 IF N>90 THEN LET N=N-26
370 IF N<65 THEN LET N=N+26
380 LET N$=N$+CHR$(N):NEXT I
390 LET M$=N$:RETURN
400 PRINT "QUAL E A MENSAGEM A SER ";C$
410 INPUT M$:RETURN
420 LET N$="":FOR I=LEN(M$) TO 1 STEP -1
▼ 430 LET N$=N$+MID$(M$,I,1):NEXT I
440 LET M$=N$:RETURN
450 LET N$="":LET L=LEN(M$)
460 FOR I=1 TO LEN(M$)-1 STEP 2
▼ 470 LET N$=N$+MID$(M$,I+1,1)
▼ 480 LET N$=N$+MID$(M$,I,1)
490 NEXT I:LET M$=N$:RETURN

```

Os espões em geral decodificam suas mensagens antes de enviá-las, para ver se estão corretas.



270: Vai para a sub-rotina que desloca as letras. Suprime a primeira letra.

310-390: SUB-ROTINA PARA DESLOCAR

310: Coloca uma "string" vazia em N\$.

320: Começo de um "loop" que remove letras de M\$ uma por uma.

330: Obtém o número de código do computador para a letra.

340: Se é um espaço, não faz nada.

350: Soma o valor de X ao código da letra.

360-370: Ajusta o número de código se for maior que o código de Z ou menor que o código de A.

380: Acrescenta uma letra a N\$ e volta ao início do "loop".

390: Depois que todas as letras foram deslocadas, a nova mensagem é colocada em M\$.

400-410: Sub-rotina que recebe uma mensagem.

420-440: Sub-rotina que inverte a mensagem.

450-490: Sub-rotina que troca as letras duas a duas.

Modificações

● 10 Substitua CLS por HOME
 ▲ 10 Substitua CLS por PRINT CHR\$(147)
 ◆ 130 LET X=INT(RND*25)+1
 ▲ 130 LET X=INT(RND(1)*25+1)
 ▼ 230 LET K\$=M\$(1)
 ▼ 240 LET M\$=M\$(2 TO)

▼ 250 LET X=CODE(K\$)-64
 ▼ 270 GOSUB 310:LET M\$=M\$(2 TO)
 ▼ 330 LET Q\$=M\$(1):LET N=CODE(Q\$)
 ▼ 430 LET N\$=N\$+M\$(1):NEXT I
 ▼ 470 LET N\$=N\$+M\$(1+1)
 ▼ 480 LET N\$=N\$+M\$(1)



XSHV OKTITT LMIHKH
IFT NAHXGYM

CJKS ZNYB
KS CYCKWPF

CJNYFSYEY YN
DYEPC

ZKDD TZC
AGLCLAPDL QU

Você é capaz de descobrir o que eles estão dizendo?



Encontro Marcado

Sua missão é complicada, de modo que você deve ler estas instruções com atenção.

Você precisa apanhar uma maleta em um escaninho da estação, entregá-la ao seu contato e voltar ao aeroporto a tempo de pegar o último avião (o computador lhe dirá a hora).

O computador lhe informa onde estará o contato e a que horas. Você tem que deixar uma mensagem nesse lugar, antes da hora marcada, dizendo onde e quando o encontrará para entregar-lhe a maleta.

Você precisa descobrir a senha antes de encontrar o contato e não pode chegar atrasado mais que 15 minutos.

Antes de pegar a maleta, você precisa encontrar a chave do escaninho e também descobrir o seu número. Infelizmente, a chave se encontra nas mãos de agentes inimigos, que estão hospedados no Hotel. Você precisa encontrar um espião inimigo e segui-lo, na esperança de que deixe cair a chave (torcendo, naturalmente, para que ele não o veja).

O mapa mostra os lugares aonde você pode ir e a lista abaixo mostra as palavras que você pode usar no jogo.

Agora coma o livro.

Talvez você queira anotar a senha e o número do escaninho (em código, naturalmente).

Palavras que podem ser usadas

HORA — Para saber que horas são.

IR — A qualquer lugar do mapa.


DIZER — A senha.

EXAMINAR — Qualquer coisa.
(Examinar a chave para saber o número.)

LER — Uma mensagem.

ABRIR — O escaninho.

SEGUIR — Um espião inimigo.



ESPERAR — Um certo tempo.

DEIXAR — Uma mensagem.


PROCURAR — Em qualquer lugar (para achar a chave).

LEMBRAR — Para saber a hora e local do encontro.

Você também pode usar qualquer dos nomes que aparecem no mapa em letras maiúsculas.

Se quiser, você pode escolher outras senhas.

Você também pode mudar os nomes dos lugares.



Se o seu micro
é da família
TRS-80, entre
com CLEAR
300 antes de
começar o jogo.

Como o programa funciona

10: Vai para o final do programa para ler os dados e inicializar as variáveis.

20: P, Y e X assinalam a posição do jogador. EP é a posição do inimigo.

50: BS é para as mensagens do computador para o jogador.

80: Aumenta NM (número de movimentos).

130: NE conta há quanto tempo o jogador está no mesmo lugar que um agente inimigo.

170: Calcula o tempo (H = horas, M = minutos).

```

10 GOSUB 1010:GOSUB 1130
20 LET P=1:LET Y=0:LET X=1:LET EP=10
30 LET U=0:LET NM=0
40 LET SS=""
50 LET BS=""
60 CLS:PRINT:PRINT "ENCONTRO MARCADO"
70 PRINT "*****"
80 LET NM=NM+1
90 PRINT:PRINT
100 PRINT BS:PRINT
110 PRINT:PRINT "VOCE ESTA N";
120 PRINT R$(P)
130 IF EP=P THEN PRINT "O AGENTE INIMIGO ESTA AQUI":LET NE=NE+1
140 IF EP<>P THEN LET NE=0
150 IF P=HP THEN PRINT "HA UMA MENSAGEM PARA VOCE"
160 LET F(3)=0
170 LET T1=H+M/100
180 IF F(4)=1 AND MID$(R$(P),3,LEN(R$(P)))=SS AND U<T1 AND (H+.15>T1) THEN PRINT TS:LET F(3)=1

```

```

190 IF P=1 AND H<FH AND F(7)=1 THEN GOTO 890
200 PRINT:PRINT:PRINT "E AGORA"
210 LET BS=""
220 INPUT IS
▼ 230 LET V=0:FOR I=1 TO 11
240 IF IS=VS(I) THEN LET V=I
250 NEXT I
◊ 260 IF V=0 THEN LET V=12
270 IF NE=3 AND RAND(10)>3 AND V<1 THEN LET BS="O AGENTE INIMIGO VIU VOCE!":GOTO 60
280 IF NE=5 THEN PRINT "VOCE FOI CAPTURADO!":STOP
290 ON V GOSUB 360,420,490,540,570,640,710,730,780,810,820,870
300 LET M=M+DT
310 IF M>59 THEN LET M=M-60:LET H=H+1:GOTO 310
320 IF F(2)=1 AND H>CH THEN LET F(4)=1
330 IF H=FH THEN GOTO 880
◊ 340 IF RAND(10)>9 THEN LET EP=10
350 GOTO 60
360 PRINT:PRINT "PARA ONDE?":INPUT NS
▼ 370 LET NP=0:FOR I=1 TO 20
380 IF NS=MID$(RS(I),3,LEN$(RS(I))) THEN LET NP=I
390 NEXT I:IF NP=0 THEN GOTO 360
400 GOSUB 950
410 LET P=NP:RETURN
420 LET DT=5
430 PRINT:PRINT "DIZER O QUE?":INPUT OS
440 IF EP=P THEN LET BS="VOCE ATARAIU O AGENTE INIMIGO!":RETURN
450 IF F(3)=0 THEN LET BS="NAO HA NINGUEM PARA OUVI-LO":RETURN
460 IF OS<PS THEN LET BS="SEU CONTATO IGNOROU-O!":RETURN
470 IF F(6)=1 THEN LET BS="VOCE FEZ CONTATO - ELE LEVOU A MALETA!":LET F(7)=1:RETURN
480 IF F(6)=0 THEN LET BS="VOCE FEZ CONTATO, MAS NAO TEM A MALETA!":RETURN
490 LET DT=5
500 PRINT:PRINT "O QUE VOCE QUER EXAMINAR?":INPUT QS
510 IF QS="CHAVE" AND F(5)=1 THEN LET BS="UM NUMERO : "+STR$(NL):RETURN
520 IF QS="CHAVE" THEN BS="QUE CHAVE?":RETURN
530 LET BS="NADA DE ESPECIAL!":RETURN
540 IF P<NP OR F(1)=1 THEN LET BS="NAO HA NADA PARA SER LIDO!":RETURN
550 LET BS="UMA PALAVRA : "+PS+"":RETURN
560 LET F(1)=1:RETURN
570 LET DT=5
580 IF P<16 THEN LET BS="NAO HA NADA PARA SER ABERTO":RETURN
590 IF F(5)=0 THEN LET BS="VOCE NAO TEM CHAVES":RETURN
600 PRINT:PRINT "QUAL O NUMERO DO ESCANINHO?":INPUT YN
610 IF NL<YN THEN LET BS="A CHAVE NAO SERVE":RETURN
620 LET BS="O ESCANINHO ESTA ABERTO - VOCE TEM A MALETA!":LET F(6)=1
630 RETURN
640 LET DT=5
650 IF EP<P THEN LET BS="SEGUIR QUEM?":RETURN
660 LET NP=RND(20):GOSUB 950:LET P=NP
◊ 670 IF RAND(10)>8 THEN LET P=KP
◊ 680 IF RAND(10)>7 THEN LET BS="VOCE O PERDEU DE VISTA APÓS UM CERTO TEMPO":RETURN
690 LET EP=P
700 LET BS="VOCE NAO O PERDEU DE VISTA":RETURN
710 PRINT:PRINT "QUANTOS MINUTOS?":INPUT DT
720 RETURN
730 PRINT:PRINT "ONDE QUER ENCONTRA-LO?":INPUT SS
▼ 740 PRINT:PRINT "A QUE HORAS (HH,MM)?"
750 INPUT HH,MM:U=HH*MM/100
760 IF P=CP AND T1<U AND H<CH THEN LET F(2)=1
770 LET DT=5:RETURN
780 LET BS="NAO HA NADA AQUI!":LET DT=10
790 IF P=KP THEN LET BS="VOCE ACHOU UMA CHAVE!":LET F(5)=1
800 RETURN

```

180-190: Verifica se a missão foi cumprida.

200-260: Recebe instrução do jogador e a compara com as que estão guardadas na memória. V é o número da palavra encontrada.

270-280: Verifica o valor de NE para saber se o jogador foi visto pelo inimigo.

290: Pula para uma entre várias sub-rotinas, dependendo da instrução recebida.

300: Incrementa o tempo.

330: Verifica se o tempo do jogador acabou.

340: Leva o inimigo de volta para o Hotel.

360-410: SUB-ROTINA PARA IR

420-480: SUB-ROTINA PARA DIZER

490-530: SUB-ROTINA PARA EXAMINAR

540-560: SUB-ROTINA PARA LER

570-630: SUB-ROTINA PARA ABRIR

640-700: SUB-ROTINA PARA SEGUIR

710-720: SUB-ROTINA PARA ESPERAR

730-770: SUB-ROTINA PARA DEIXAR

780-800: SUB-ROTINA PARA PROCURAR

810: SUB-ROTINA PARA DIZER A HORA



820-860: SUB-ROTINA PARA LEMBRAR

870: SUB-ROTINA PARA PALAVRAS DESCONHECIDAS

880: Mensagem para quando o tempo acaba.

890-940: Imprime mensagem e calcula a contagem de pontos para uma missão bem-sucedida. A contagem depende do número de movimentos e do tempo que resta.

950-1000: Sub-rotina para calcular a distância percorrida e o tempo gasto pelo jogador.

1010-1050: Sub-rotina para ler os dados.

1060-1120: Linhas de dados.

1130: Escolhe a hora inicial.

1140: Escolhe a hora do último vôo.

1150: Escolhe a hora para o contato apanhar a mensagem.

1180-1190: Escolhe os locais da mensagem, da chave, do inimigo e onde o contato vai apanhar a mensagem.

1200-1210: Escolhe a senha.

1220: Escolhe o número do escaninho.


1230-1320: Imprime a apresentação do jogo.

```

810 LET DT=0:LET BS="SÃO "+STR$(H)+" HORAS E "+STR$(M)+" MINUTOS":RETURN
820 LET DT=5
830 IF U=0 THEN GOTO 860
840 PRINT:PRINT "O LUGAR DE ENCONTRO É"
850 PRINT SS: AS "":MM: HORAS E "":MM: MINUTOS
860 GOSUB 1300:RETURN
870 LET DT=0:LET BS="HEIN?":RETURN
880 PRINT "TARDE DEMAIS ":STOP
890 PRINT:PRINT "MUITO BEM. SUA MISSÃO FOI UM SUCESSO!"
900 LET TL=(FH-H)*60-M
910 LET S=INT((20/NM+TL/120)*50)
920 PRINT:PRINT "SUA NOTA"
930 PRINT "FOI "S
940 STOP
950 LET NY=INT((NP-1)/5)
960 LET NX=NP-S*NY
970 LET DX=ABS(X-NX):LET DY=ABS(Y-NY)
980 LET X=NX:LET Y=NY
990 LET D=SQ(DX*DX+DY*DY)
1000 LET DT=INT(S*DT):RETURN
1010 DIM RS(20),VS(11),F(7)
1020 FOR I=1 TO 20:READ RS(I)
1030 NEXT I
1040 FOR I=1 TO 11:READ VS(I):NEXT I
1050 RETURN
1060 DATA "O AEROPORTO","A PARADA DE ÔNIBUS","A PONTE","O RIO","A IGREJA"
1070 DATA "O PARQUE","O RESTAURANTE","O BANCO","O CINEMA","O HOTEL"
1080 DATA "A ESCOLA","A PRACA","O CORREIO","A DELEGACIA","O CIRCO"
1090 DATA "A ESTACAO","A PREFEITURA","A EMBAIXADA","O JARDIM","O CASTELO"
1100 DATA "IR","DIZER","EXAMINAR","LER","ABRIR","SEGUIR","ESPERAR","DEIXAR","PROCURAR"
1110 DATA "HORA","LEMBRAR"
1120 DATA "APITO","AIPIM","HOSTARDA","CANGURU","MOSQUITO","BANDOLIM"
1130 LET H=RND(2)+B:LET M=0
1140 LET FH=RND(2)+14
1150 LET CH=RND(2)+H
1160 CLS:PRINT:PRINT
1170 LET NE=0:LET TS="O CONTATO ESTÁ AQUI"
1180 LET MP=RND(18):LET KP=RND(18)
1190 LET EP=RND(18):LET CP=RND(18)
1200 FOR I=1 TO RND(6)
1210 READ PS:NEXT I
1220 LET NL=RND(900)+99
1230 PRINT "ENCONTRO MARCADO"
1240 PRINT "=====":PRINT
1250 PRINT "APANHE MENSAGEM"
1260 PRINT "N":RS(MP)
1270 PRINT "O CONTATO ESTARÁ"
1280 PRINT "N":RS(CP)
1290 PRINT "AS "CH:" HORAS"
1300 PRINT "O ÚLTIMO VOO PARTE"
1310 PRINT "AS "FH:" HORAS"
1320 PRINT:PRINT"APORTE RETURN PARA CONTINUAR"
1330 INPUT QS:RETURN

```

As sub-rotinas do programa estão nas linhas de 360 a 870.



Nas linhas de 1010 a 1320 está a chamada "inicialização" do programa.

Modificações

- 60,1160 Substitua CLS por HOME
- ▲ 60,1160 Substitua CLS por PRINT CHR\$(147)
- ▼ 225 IF LEN(1\$)<7 THEN LET 1\$=1\$+" ":GOTO 225
- ▼ 290 GOSUB 360*(V=1)+420*(V=2)+490*(V=3)+540*(V=4)+570*(V=5)+640*(V=6)+710*(V=7)+730*(V=8)+780*(V=9)+810*(V=10)+820*(V=11)+870*(V=12)
- ▼ 365 IF LEN(NS)<14 THEN LET NS=NS+" ":GOTO 365
- ▼ 380 LET Z\$=RS(1):IF NS=Z\$(3 TO LEN(Z\$)) THEN LET NP=1
- ▼ 735 IF LEN(SS)<14 THEN LET SS=SS+" ":GOTO 735
- ▼ 1010 DIM RS(29,14):DIM VS(11,7):DIM F(7)
- ▼ 270,340,660,670,680,1130,1140,1150,1180,1190,1200,1220 Substitua RND(n) por INT(RND*n+1)
- ▲ 270,340,660,670,680,1130,1140,1150,1180,1190,1200,1220 Substitua RND(n) por INT(RND(1)*n+1)

Veja se consegue descobrir para que servem todas as linhas das sub-rotinas.

As variáveis F(1), F(2) etc. são "bandeiras". F(5), por exemplo, é 1 se você está com a chave e 0 se não está. Veja se descobre para que servem as outras bandeiras.

Código Morse

Se você quer ser um bom espião, terá que aprender a enviar, receber e também, naturalmente, interceptar mensagens em Código Morse. Este programa o ajudará a praticar. Se você não conhece o Código Morse, terá que preparar uma tabela com as letras e seus equivalentes em código. Para isso, pode usar as linhas 400 a 450 do programa, que mostram os sinais de Código Morse de todas as letras em ordem alfabética.

Depois que aprender o código, não precisará mais da tabela.



A velocidade é controlada pela linha 30. Quanto menor o número, mais rápido o programa.



Você pode acrescentar uma rotina de contagem de pontos, semelhante à do programa Espiobot.

Você também pode usar sons para acompanhar ou substituir a estrela. Use um som curto para os pontos e um som 3 vezes mais longo para os traços.



Como usar o programa

No Código Morse, cada letra é representada por uma série de sinais longos e curtos. Este programa usa uma estrela que pisca. Ele transmite o código de uma letra e pergunta a você qual foi essa letra.

Como funciona

10: Vai para a sub-rotina que lê os dados.

30: Estabelece a velocidade.

50-80: Pergunta qual o nível que você quer e calcula a velocidade do jogo dependendo da resposta.

110: Pausa

120: Vai para a sub-rotina que escolhe uma letra ao acaso e a guarda em Q\$.

130: Determina o Código Morse para a letra e o guarda em F\$.

140: Vai para a sub-rotina que imprime a estrela.

160-170: Recebe a sua resposta.

180-190: Verifica se a resposta está correta e imprime uma mensagem apropriada.

200-210: Faz uma pausa e volta para a letra seguinte.

220-250: Examina todos os elementos de F\$, fazendo K igual a 1 para os pontos e igual a 3 para os traços. (K determina o tempo que a estrela permanece na tela.)

260-270: Vai para a sub-rotina que imprime a estrela. Volta, faz K igual a 1 e volta para a sub-rotina na linha 340 para "imprimir" um espaço. O objetivo é deixar um espaço entre os pontos e traços.

280: Volta para o ponto ou traço seguinte.

310-320: Sub-rotina para escolher uma letra ao acaso.

330: Imprime a estrela.

340: Espera por um período de tempo que depende de K.

350: Imprime um espaço no lugar da estrela.

370-390: Sub-rotina que lê os dados.

400-450: Linhas de dados.

```

10 GOSUB 370
▲ 20 CLS
30 LET S=30
40 PRINT:PRINT "CODIGO MORSE"
50 PRINT:PRINT "QUAL O NIVEL ?"
60 PRINT:PRINT "(1=RAPIDO)"
70 PRINT:PRINT "(5=LENTO)"
80 INPUT P:LET P=P*S
▲ 90 CLS
100 PRINT:PRINT "PREPARE-SE"
110 FOR T=1 TO 20:S:NEXT T
120 GOSUB 310
▼ 130 LET F$=M$(ASC(Q$)-64)
140 GOSUB 220
▲ 150 CLS:PRINT
160 PRINT "ESCREVA A RESPOSTA"
170 INPUT X$
180 IF X$=Q$ THEN PRINT "CERTO"
190 IF X$(>)Q$ THEN PRINT "NAO. A LETRA ERA ";Q$
200 FOR T=1 TO 30:S:NEXT T
210 GOTO 90
220 FOR J=1 TO LEN(F$)
▼ 230 LET W$=MID$(F$,J,1)
240 IF W$="." THEN LET K=1
250 IF W$="-" THEN LET K=3
260 GOSUB 330:LET K=1
270 GOSUB 340
280 NEXT J
290 RETURN
300 PRINT
▼ 310 LET Q$=CHR$(INT(RND(0)*26+65))
320 RETURN
▼ 330 PRINT 240,"*"
340 FOR T=1 TO P*K:NEXT T
▼ 350 PRINT 240," "
360 RETURN
▼ 370 DIM M$(26)
380 FOR I=1 TO 26:READ M$(I):NEXT I
390 RETURN
400 DATA "A",".-","-.-.-"
410 DATA "B","-","-.-","-.-.-"
420 DATA "C","-","-","-.-","-.-.-"
430 DATA "D","-","-","-","-.-"
440 DATA "E","-","-.-"
450 DATA "F","-","-","-","-.-"

```

Modificações

- 20,90,150 Substitua CLS por HOME
- ▲ 20,90,150 Substitua CLS por PRINT CHR\$(147)
- ▼ 130 LET F\$=M\$(CODE(Q\$)-64)
- ▼ 230 LET W\$=F\$(J)
- ▼ 235 IF W\$="." THEN GOTO 280
- ▼ 310 LET Q\$=CHR\$(INT(RND*26+65))
- ▲ 310 LET Q\$=CHR\$(INT(RND(1)*26+65))
- ▼ 330 PRINT AT 10,10;"*"
- 330 VTB(10):PRINT TAB(10);" "
- ▲ 330 PRINT CHR\$(19):FOR I=1 TO 10:PRINT:PRINT TAB(10);" "
- ▼ 330 LOCATE 10,10:PRINT "*"
 - 350 PRINT AT 10,10;" "
 - 350 VTB(10):PRINT TAB(10);" "
 - ▲ 350 PRINT CHR\$(19):FOR I=1 TO 10:PRINT:PRINT TAB(10);" "
 - ▼ 350 LOCATE 10,10:PRINT " "
- ▼ 370 DIM M\$(26,4)

Respostas dos problemas

Aqui estão as respostas de alguns dos problemas apresentados neste livro. Pode ser que sejam diferentes das suas. Se os seus programas funcionarem corretamente, tudo bem. Entretanto, verifique se suas respostas são tão simples e objetivas quanto as do livro.

Olhos de Espião

Aqui está como mostrar o recorde do momento.

```
15 LET H=0
215 PRINT "RECORD = ";H
275 IF P>H THEN LET H=P:PRINT "ESTE E O NOVO RECORD"
290 INPUT A$:IF A$="S" THEN GOTO 20
```

Holofote

Aqui está como fazer o holofote ficar aceso por mais tempo...

```
430 IF C1=TC THEN LET L=1:LET C1=0:LET TC=INT(RND(0)*10+(8-A))+10:GOSUB 300
```

...e aqui está como dar ao jogador uma segunda oportunidade.

```
15 LET NJ=0
255 LET NJ=NJ+1
257 IF NJ<2 THEN FOR T=1 TO 200:NEXT T:GOTO 30
```



Para mais de duas tentativas, aumente este número.



Este "loop" faz uma pausa para que você possa ler a mensagem. Para uma pausa maior, basta usar um número maior que 200.



Use a versão de RND apropriada para o seu micro.

NJ é uma nova variável para guardar o número de jogadas.



Espiobot

As linhas abaixo fazem o computador perguntar o nome do jogador quando este bate o recorde.

```
25 LET H$="NINGUEM"
330 Suprima esta linha
365 IF E=1 AND S>H THEN LET H=S:GOSUB 410
410 PRINT "ESTE E O NOVO RECORDE!"
420 PRINT "ESCREVA SEU NOME, POR FAVOR"
430 INPUT H$:RETURN
```

Teste para Espiões

Aqui está como mudar o número de posições de 10 para 15

```
10 DIM N(15):DIM N$(5)
70,160,200,460 Mude o 10 nessas linhas para 15
```

Nota para os usuários do MSX: Em todos os programas, acrescente a linha:

```
5 R=RND(-TIME)
```

Nota para os usuários do PC: Em todos os programas, acrescente a linha:

```
5 RANDOMIZE VAL(RIGHT$(TIME$,2))
```

Nota especial para os usuários do TK-2000

Se o seu micro é um TK-2000, use as modificações indicadas para a família Apple, com uma única exceção: se a modificação for do tipo

nnn a\$="":IF PEEK(-16384))127 THEN GET a\$
onde nnn é o número da linha e a\$ é o nome de uma variável, substitua esta linha por:

nnn a\$="":CALL -16397:IF PEEK(-16385))0 THEN GET a\$
e acrescente as seguintes linhas:

```
1 DATA 169,255,141,0,192,77
```

```
2 DATA 16,192,141,255,191,96
```

```
3 FOR I=-16397 TO -16386
```

```
4 READ A:POKE I,A:NEXT I
```

Guias Práticos de Microcomputadores



TUDO O QUE VOCÊ PRECISA SABER SOBRE MICROCOMPUTADORES

e as maravilhas que eles podem fazer



NOVOS * COLORIDOS * FASCINANTES